
QU’EST-CE QU’UNE USERFORM ? 
U ne U serForm  est une fenêtre ou une boîte de dialogue destinée à servir d’interface utilisateur 
personnalisée dans une application. Dans l’interface utilisateur W ord, il existe déjà de nom breuses 
boîtes de dialogue. Voici par exemple la boîte de dialogue Fichier/Imprimer de Word 2003 : 

 

On y trouve de nombreux boutons, des listes déroulantes, des cases à cocher, une zone de texte à 
rem plir, du texte et des im ages d’explication…  Tous ces objets d’appellent des contrôles. Les 
contrôles sont disposés sur la feuille qui sert de fenêtre à la boîte de dialogue. 



V oici m aintenant une autre boîte de dialogue pouvant s’ouvrir dans W ord  : 

 

Inutile de chercher celle-ci dans votre Word, c’est une UserForm ! Si elle ressemble un peu à la 
précédente avec des contrôles similaires, elle com m ande à l’évidence des fonctions inhabituelles 
pour Word : ajouter à l’im pression un filigrane non présent dans le docum ent, gérer un com pteur 
des copies imprimées ou tenir un journal des impressions. 

Cet exemple permet de comprendre qu’une U serForm  est un objet complexe qui comporte toujours 
les trois éléments suivants : 

 La feuille : c’est la fenêtre qui s’affiche.  
 Les contrôles : ce sont tous les objets présents dans la feuille et pouvant interagir avec 

l’utilisateur. 
 Le code : c’est ce qui décide des actions déclenchées par l’utilisateur. 
La feuille est désignée par le nom  de l’objet UserForm lui-même. Les contrôles sont des objets 
incorporés dans la UserForm : ils font partie de ses propriétés. Le code enfin, sans lequel une 
UserForm n’aurait aucune action est contenu également dans la UserForm au m êm e titre qu’un 
module dans un projet. 

U n m ot encore sur le code. L’essentiel en est la gestion des événements. Une fois ouverte et 
affichée, la UserForm doit réagir à un certains nom bres d’actions de l’utilisateur. A insi cliquer sur le 
bouton O K est un événem ent. Par cette action on s’attend que le code valide toutes les données 
entrées et réalise le propos de la boîte de dialogue comme ici lancer une impression au moment de 
refermer la UserForm elle-même. 

EXEMPLE DE USERFORM POUR COMPLETER UN DOCUMENT 
Nous choisirons un exemple simple de Userform utilisé comme formulaire pour compléter un 
document. Supposons un modèle de document ainsi constitué : 



 

N ous voulons qu’à la création d’un nouveau docum ent basé sur ce m odèle s’ouvre un form ulaire 
facilitant la saisie rapide du texte placé entre guillemets « Titre », « Question », « Répondeur », 
« Date », lesquels sont en réalité des signets dans le document. 

CREER UNE NOUVELLE USERFORM 
La création d’une UserForm est très simple dans VBE l’éditeur visual-basic : 

 Créer un document vierge Fichier / Nouveau  
 Ouvrir VBE par Outils / Macro / Visual Basic Editor ou Alt+F11 
 D ans la fenêtre de l’explorateur de projets, sélectionner le projet du document créé 
 Menu Insertion / Userform 
 

 

C ’est tout : le document contient désormais une UserForm. Son nom est UserForm1 par défaut. 
Affichons au besoin les propriétés de UserForm1 avec F4. Nous pouvons redimensionner UserForm1 



et modifier ses propriétés. Nous pouvons changer son nom et changer son titre (caption). Gardons le 
nom UserForm1 mais changeons le titre en Formulaire. 

Avant de commencer à remplir notre UserForm1, jetons un coup d’œ il sur son code en cliquant sur 
le bouton ‘A fficher le code’ de la fenêtre Projet ou avec F7 pour constater que ce code est vide 
pour l’instant. Revenons à l’objet U serForm  avec Maj+F7. 

PLACER DES CONTROLES 
C ’est un sim ple petit jeu de construction graphique ! Il suffit d’utiliser la boîte à outils des 
contrôles : Affichage / Boîte à outils, de choisir parmi les contrôles disponibles et de les dessiner 
sur la UserForm.  

 
La boîte à outils contrôles 

Affichons aussi la barre d’outils U serForm  qui contient de quoi faciliter la mise en place des 
contrôles. Pour chaque contrôle, il convient dès sa création d’en définir les propriétés initiales en 
utilisant très largement la fenêtre Propriétés. Si vous hésitez à modifier une de ces propriétés, 
sélectionnez-là et voyez avec F1 dans l’A ide VBA à quoi elle correspond. À  tout m om ent il sera 
possible de redéfinir les propriétés d’un contrôle.  

Notre exemple comportera les contrôles suivants : 

 Une zone de texte 
 Une liste déroulante ComboBox 
 Deux radio-boutons s’excluant l’un l’autre dans un contrôle Cadre par leur propriété  
 Un contrôle calendrier. Ce dernier n’est pas disponible par défaut dans la boîte à outils. Il nous 

faut le chercher par clic droit dans la boîte à outils / Contrôles supplémentaires / Contrôle 
Calendrier 

 Deux boutons : l’un pour annuler l’autre pour valider (O K) 
 Enfin trois Intitulés ou labels, simples contrôles passifs pour la lisibilité du formulaire. 
 



 
La fenêtre des Propriétés. 

N otez ici la propriété G roupN am e com m une aux deux boutons d’option. En définissant le m êm e 
nom  de groupe, les deux options (M /F) s’excluent mutuellement sans avoir besoin de code. 

Au total donc 11 contrôles sur notre UserForm une fois construite : 



 

Il est im portant de bien prévoir et disposer ses contrôles avant de passer à l’étape d’écriture du 
code. 

LE CODE DE LA USERFORM 
Reste maintenant à écrire tout le code VBA pour notre UserForm ! Plus la UserForm disposera de 
contrôles plus le code sera com pliqué. À l’aide de notre petit exem ple nous en distinguerons 4 
parties classiques : 

AFFICHER ET FERMER LA USERFORM 
La U serform  doit d’abord être chargé en m ém oire avec tous ses objets. C ’est le rôle de la 
commande Load. Par la suite notre objet UserForm désigné par son nom pourra être affichée par sa 
méthode Show et refermée par sa méthode Hide. V BA accepte d’exécuter im plicitem ent la 
commande Load si on l’oublie avant l’exécution d’un Show. 

D ans notre exem ple le chargem ent et l’affichage de notre U serForm  est exécuté dans une m acro 
autom atique à chaque création d’un nouveau docum ent à partir de notre m odèle. N ous y ajouterons 
la réouverture de la U serForm  pour éventuelle correction à l’aide d’un bouton -macro. Ce bouton-
macro exécute la macro Formulaire qui se contente d’exécuter à nouveau Show. 

INITIALISER LA USERFORM 
Il s’agit dans cette phase de renseigner les diverses propriétés de la U serForm  et de ses contrôles 
avant qu’elle ne s’ouvre. D ans notre exem ple il faut notam m ent renseigner la liste des répondeurs 
dans la combobox et ajuster le contrôle calendrier à la date du jour. 

Le code d’initialisation d’une UserForm peut être très complexe surtout s’il faut rechercher des 
données dans le document lui-même ou dans des fichiers externes. On aurait pu imaginer par 



exemple que la liste des répondeurs de la combobox soit à chercher dans une feuille excel, laquelle 
contiendrait également pour chacun le délai moyen de réponse à appliquer au calendrier. 

Toute cette phase d’initialisation peut être placée dans le code d’appel entre Load et Show ou 
m ieux dans le code de l’événem ent Initialize de la UserForm elle-même. Cet événement est lancé à 
chaque Load. Notez que Show ne provoque pas un événement Initialize. 

LE CODE DES CONTROLES 
C ’est tout le code événem entiel dictant com m ent doit réagir chaque contrôle aux actions de 
l’utilisateur. Les événements possibles sont très nombreux pour chaque contrôle mais il suffit de 
coder uniquem ent ceux qui doivent déclencher une réponse. U n bon nom bre de contrôles n’a pas 
besoin de code. Ce sont surtout les clics sur les boutons qu’il faut coder. 

O n accède facilem ent au code d’un contrôle par clic droit / Code  : la fenêtre de code affiche en 
haut le contrôle et les divers événem ents qu’il est possible de coder. 

Le bouton Annuler referme la UserForm par un Hide sans rien faire d’autre. 

C ’est le bouton de validation qui est le plus im portant. L’événem ent Click sur le bouton OK doit en 
effet 1) collecter les données de la UserForm 2) vérifier leur cohérence en refusant au besoin de 
refermer la UserForm 3) exécuter le propos de la UserForm et enfin 4) fermer la UserForm. 

L’étape de vérification des données peut aussi s’effectuer au niveau d’un contrôle par exemple 
avec son événement Change chaque fois que sa valeur est modifiée. 

Dans notre exemple après vérification de la zone de texte et du calendrier, la validation de la saisie 
doit provoquer l’insertion des données dans le document. 

LE CODE DE L’APPLICATION 
Ici on range le code spécialisé, les macros, les procédures et fonctions du projet pour lequel on a 
généralement développé précisément la UserForm. Ce code est généralement placé dans un ou 
plusieurs modules de façon à ne pas surcharger le code événementiel de la Userform qui y fera 
appel. 

Dans notre exemple nous plaçons un module Signets chargé de traiter la mise à jour des signets dans 
un docum ent. En effet le sim ple rem placem ent du texte d’un signet par un autre efface le signet en 
vba. La procédure RemplirSignet(Signet, Texte) rem place le contenu d’un Signet par Texte sans 
détruire ce signet, que ce dernier soit initialement vide ou non. Cette procédure est utilisée par 
notre UserForm pour remplir le document. 

Placé dans le même projet ce code est appelé par le code événementiel des contrôles de la 
Userform. 

LE CODE VBA FINAL DE L’EXEMPLE 
Tout l’exem ple est téléchargeable dans le m odèle ci-joint ‘Exemple Userform.dot’ qui 
s’utilise com m e n’im porte quel m odèle de docum ent. En allant dans VBE avec Alt+F11 on y 
trouvera la UserForm décrite plus haut ainsi que tout le code reproduit ci après : 

Private Sub Document_New() 
' Lancement automatique à chaque nouveau document 
Load UserForm1 
UserForm1.Show 



End Sub 
 
 
Public Sub Formulaire() 
' Réaffiche la UserForm 
UserForm1.Show 
End Sub 
 
 
Option Explicit 
 
Private Sub UserForm_Initialize() 
' Lors du chargement de la UserForm : opérations d'initialisations 
 
' 1 - Remplissage de la liste déroulante 
With Me.ComboBox1 
    .AddItem "Circé" 
    .AddItem "Geo" 
    .AddItem "André" 
    .AddItem "Guy" 
    .AddItem "Anacoluthe" 
End With 
 
' 2 - Initialisation du calendrier: aujourd'hui + 2 jours 
Me.Calendar1.Value = Date + 2 
 
' 3 - Initialisation du focus 
Me.TextBox1.SetFocus 
 
End Sub 
 
 
Private Sub CommandButton1_Click() 
' = Bouton Annuler 
' Lui affecter la propriété Cancel à True 
' pour bénéficier de la touche Echap (ESC) 
Me.Hide 
End Sub 
 
 
Private Sub CommandButton2_Click() 
' = Bouton OK 
 
' Les données de la UserForm : 
Dim T As String ' Titre 
Dim Q As String ' Question 
Dim R As String ' Répondeur 
Dim D As String ' Date 



 
' 1 - Récupérer les données 
 
If Me.OptionButton1 = True Then 
      T = "Cher abonné" 
Else: T = "Chère abonnée" 
End If 
 
Q = Me.TextBox1.Text 
 
R = Me.ComboBox1.Value 
 
D = Str(Me.Calendar1.Day) 
If D = " 1" Then D = " 1er" 
D = D & " " & Format(Me.Calendar1.Value, "MMMM") 
 
 
' 2 - Vérifier la cohérence des données 
 
If Me.TextBox1.Value = "" Then 
   MsgBox "Il manque la question !", vbExclamation, "Erreur" 
   Exit Sub 
End If 
 
If Me.Calendar1.Value < Date Then 
   MsgBox "On ne peut répondre avant aujourd'hui !", vbExclamation, 
"Erreur" 
   Me.Calendar1.Value = Date + 2 
   Exit Sub 
End If 
 
' 3 - Placer les données dans le document 
 
RemplirSignet "Titre", T 
RemplirSignet "Question", Q 
RemplirSignet "Répondeur", R 
RemplirSignet "Date", D 
 
ActiveDocument.Fields.Update ' màj des champs pour le renvoi sur 
Titre 
 
' 4 - Fermer la Userform 
 
Me.Hide 
 
End Sub 
 
 



 
' ******************************* 
' *** Utilisation des signets *** 
' ******************************* 
 
Public Sub RemplirSignet(S As String, T As String) 
' Remplit le signet S avec le texte T sans détruire S 
On Error GoTo rien 
Dim Place As Long 
Place = ActiveDocument.Bookmarks(S).Range.Start 
ActiveDocument.Bookmarks(S).Range.Text = T 
ActiveDocument.Bookmarks.Add Name:=S, _ 
    Range:=ActiveDocument.Range(Place, Place + Len(T)) 
rien: 
End Sub 

